Equations and FORTRAN Program for the Canadian Forest Fire Weather Index System

Canada

Government of Canada
Canadian Forestry Service

Gouvernement du Canada
Service canadien des forêts

THE CANADIAN FORESTRY SERVICE
GOVERNMENT OF CANADA

The Canadian Forestry Service is the principal source of federal expertise in forestry. Its general objective is to promote the wise management and use of Canada's forest resources for the economic, social, and environmental benefit of Canadians.

The following are the main functions of the CFS:

1. Coordination of federal policies, for the promotion of better resource management and forest industry development.
2. Provision of scientific and technological leadership in forestry through research and development.
3. Provision and analysis of national and international statistics and information as a basis for policy formulation.
4. Development and certification of codes and standards for wood product performance.
5. Protection of Canada's forests from foreign pests.
6. Fostering the potential use of the forest resource for energy.
7. Contributing to the environmental objectives of the government of Canada.

A number of federal agencies are involved in forestry programs and a Federal Forestry Sector Strategy Committee has been established to coordinate federal forestry activities. The Canadian Forestry Service has been designated the lead agency role.

The Canadian Forestry Service is comprised of a Headquarters Unit, six Forest Research Centres and two National Institutes. The Forest Research Centres are responsive to regional priorities and maintain close liaison with the respective provincial government forestry departments and other clients. They also participate in, and frequently lead, national programs. The National Institutes provide the focus for programs of national scope.

Equations and FORTRAN Program for the Canadian Forest Fire Weather Index System

C.E. Van Wagner and T.L. Pickett
Petawawa National Forestry Institute

Canadian Forestry Service
Government of Canada
Forestry Technical Report 33
Ottawa 1985

ISBN 0-662-13906-2
Cat. No. Fo64-33/1985E

Additional copies are available at no charge from:

Environment Canada
Distribution Centre
151 Jean-Proulx
Ottawa, Ont.
K1A 1C7

Cette publication est aussi disponible en français sous le titre *Équations et programme FORTRAN de l'Indice Forêt-Météo de la méthode canadienne.*

CONTENTS

ABSTRACT

Improved official equations are presented for the 1984 version of the Canadian Forest Fire Weather Index System. The most recent mathematical refinements serve to further rationalize the Fine Fuel Moisture Code and render it more compatible with other developments in the Canadian Forest Fire Danger Rating System. The effect of these changes is so slight that no problems are anticipated in converting from the previous version to this new one. Also given is a FORTRAN program intended as a standard for processing the equations in their most accurate mathematical form.

RÉSUMÉ

Ce rapport présente, en complément au système de l'Indice canadien forêt-météo, des équations connues qu'on a retouchées. Ces précisions mathématiques veulent rendre l'indice du combustible léger plus rationnel et davantage compatible avec les nouveaux développements que connaît la Méthode canadienne d'évaluation des dangers d'incendie de forêt. Il s'agit de changements mineurs qui ne devraient pas faire problème advenant qu'une équation de la version antérieure soit à convertir. Le programme FORTRAN se veut une base pour traiter des équations dans le plus pur langage mathématique.

INTRODUCTION

The present report is a companion to the fourth edition of the Tables for the Canadian Forest Fire Weather Index System (Canadian Forestry Service 1984). The equations on which this new edition is based are presented here. Also included is a computer program incorporating the mathematical routines leading to the Fire Weather Index (FWI). It is written in FORTRAN 77 to run on a DEC PDP-11/44. With proper modification it could be handled by many smaller computers.

The report is a replacement of Information Report PS-X-58 (Van Wagner and Pickett 1975[1]), which listed equations for the 1976 version of the Fire Weather Index (FWI) System. The development and structure of the FWI System were described originally by Van Wagner (1974) in Canadian Forestry Service Publication 1333, but the mathematics in that paper are partly obsolete. A revised version of Publication 1333 will soon be available.

The 1976 version of the FWI System listed the following mathematical changes from the original 1970 version:

1. Change in equation giving the equilibrium moisture content (EMC) during a wetting cycle in the Fine Fuel Moisture Code (FFMC), to eliminate an anomaly at very high relative humidity (RH);
2. Replacement of the original temperature correction in the FFMC with separate temperature effects on EMC and drying rate;
3. Increase in the amount of rain discarded in the Drought Code (DC) from 1.524 mm to 2.8 mm;
4. Change in the buildup function, f(D), at Buildup Index (BUI) values over 80, in order to cause the FWI to level off in extreme drought.

Note: Further acronyms and symbols are defined in accompanying lists.

The additional mathematical changes in the present version further rationalize the structure of the FFMC. This code has worked adequately, but new developments in the Canadian Forest Fire Danger Rating System (CFFDRS), of which the FWI System is a part, create a need for a) realistic conversions from code value to actual fuel moisture, b) a common basis for the standard daily FFMC and its round-the-clock hourly version (Van Wagner 1977) and, c) ready modification for special purposes. Accordingly, the new FFMC incorporates the following alterations:

1. A new moisture scale, the FF-scale, with a higher maximum possible moisture content (250% instead of 101%);

2. A new rain-effect equation that operates on moisture content rather than on code value;

3. An adjusted drying rate that operates on the moisture content as expressed by the new moisture scale, but results in the same rate of change in code value as before;

4. A variable wetting rate when the EMC is higher than the initial moisture content, in place of the fixed rate used previously, and;

5. An adjustment of the fine fuel moisture function, f(F), in the Initial Spread Index (ISI), to fit the new higher moisture conversion scale.

---

[1]C.E. Van Wagner is a research scientist, and T.L. Pickett is a computer programmer at the Petawawa National Forestry Institute, Chalk River, Ontario K0J 1J0.

These changes have been accomplished with very little difference in the output of the system of equations. That is, the resulting codes and indexes will be sufficiently similar for daily use of the system to continue uninterrupted and for the statistical continuity of annual records to be preserved.

As before, the computer program is set up for the calculation at one time of an entire season's output for one station only. Its primary purpose is to illustrate the programming of the equations in order to give the purest possible mathematical output, with no artificial constraints or limits other than those required by the nature of the equations. It is intended to serve as a standard of comparison by which the output of any other program may be tested.

This 1984 version (as well as the previous version) includes the computation of the Daily Severity Rating (DSR), as described by Van Wagner (1970). The DSR is a function of the FWI specifically designed for averaging, either for any desired period of time at a single location, or as an area average of any number of locations. The FWI itself, on the other hand, is not considered suitable for averaging, and should be used as its single daily value only. The DSR averaged over a whole season, for example, would be called the Seasonal Severity Rating (SSR) and could be used for comparing fire weather from year to year, or fire climate from place to place.

SYMBOLS IN THE EQUATIONS

All quantities used in the numbered equations are represented in the following list by single letters, sometimes with subscripts. The symbols are arranged in groups according to their place in the whole. All moisture contents are percentages based on dry weight. Noon refers to standard time.

WEATHER

T	– noon temperature, °C
H	– noon relative humidity, %
W	– noon wind speed, km/h
r_o	– rainfall in open, measured once daily at noon, mm
r_f	– effective rainfall, FFMC
r_e	– effective rainfall, DMC
r_d	– effective rainfall, DC

FINE FUEL MOISTURE CODE (FFMC)

m_o	– fine fuel moisture content from previous day
m_r	– fine fuel moisture content after rain
m	– fine fuel moisture content after drying
E_d	– fine fuel EMC for drying
E_w	– fine fuel EMC for wetting
k_o	– intermediate step in calculation of k_d
k_d	– log drying rate, FFMC, $\log_{10}$m/day
k_l	– intermediate step in calculation of k_w
k_w	– log wetting rate, $\log_{10}$m/day
F_o	– previous day's FFMC
F	– FFMC

DUFF MOISTURE CODE (DMC)

M_o	– duff moisture content from previous day
M_r	– duff moisture content after rain
M	– duff moisture content after drying
K	– log drying rate in DMC, $\log_{10}$M/day
L_e	– effective day length in DMC, hours
b	– slope variable in DMC rain effect
P_o	– previous day's DMC
P_r	– DMC after rain
P	– DMC

DROUGHT CODE (DC)

Q	– moisture equivalent of DC, units of 0.254 mm
Q_o	– moisture equivalent of previous day's DC
Q_r	– moisture equivalent after rain
V	– potential evapotranspiration, units of 0.254 mm water/day
L_f	– day length adjustment in DC
D_o	– previous day's DC
D_r	– DC after rain
D	– DC

FIRE BEHAVIOR INDEXES (ISI, BUI, FWI)

$f(W)$	– wind function
$f(F)$	– fine fuel moisture function
$f(D)$	– duff moisture function
R	– Initial Spread Index (ISI)
U	– Buildup Index (BUI)
B	– FWI (intermediate form)
S	– FWI (final form)

SEVERITY RATING

DSR	– Daily Severity Rating

EQUATIONS AND PROCEDURES

FINE FUEL MOISTURE CODE (FFMC)

$m_o = 147.2\,(101 - F_o)/(59.5 + F_o)$		(1)
$r_f = r_o - 0.5,$	$r_o > 0.5$	(2)
$m_r = m_o + 42.5\, r_f \left(e^{-100/(251 - m_o)}\right)\left(1 - e^{-6.93/r_f}\right),$	$m_o \leq 150$	(3a)
$m_r = m_o + 42.5\, r_f \left(e^{-100/(251 - m_o)}\right)\left(1 - e^{-6.93/r_f}\right) + 0.0015\,(m_o - 150)^2\, r_f^{0.5},$	$m_o > 150$	(3b)
$E_d = 0.942\, H^{0.679} + 11e^{(H - 100)/10} + 0.18\,(21.1 - T)\left(1 - e^{-0.115H}\right)$		(4)
$E_w = 0.618\, H^{0.753} + 10e^{(H - 100)/10} + 0.18\,(21.1 - T)\left(1 - e^{-0.115H}\right)$		(5)
$k_o = 0.424\left[1 - (H/100)^{1.7}\right] + 0.0694\, W^{0.5}\left[1 - (H/100)^8\right]$		(6a)
$k_d = k_o \times 0.581\, e^{0.0365T}$		(6b)
$k_l = 0.424\left[1 - \left(\frac{100 - H}{100}\right)^{1.7}\right] + 0.0694\, W^{0.5}\left[1 - \left(\frac{100 - H}{100}\right)^8\right]$		(7a)
$k_w = k_1 \times 0.581\, e^{0.0365T}$		(7b)
$m = E_d + (m_o - E_d) \times 10^{-k_d}$		(8)
$m = E_w - (E_w - m_o) \times 10^{-k_w}$		(9)
$F = 59.5\,(250 - m)/(147.2 + m)$		(10)

The FFMC is calculated as follows:

1. Previous day's F becomes F_o.

2. Calculate m_o from F_o by Equation 1.

3a. If $r_o > 0.5$, calculate r_f by Equation 2.
 b. Calculate m_r from r_f and m_o by Equation 3a or 3b.
 (i) If $m_o \leq 150$, use Equation 3a.
 (ii) If $m_o > 150$, use Equation 3b.
 c. Then, m_r becomes the new m_o.

4. Calculate E_d by Equation 4.

5a. If $m_o > E_d$, calculate k_d by Equations 6a and 6b.
 b. Calculate m by Equation 8.

6. If $m_o < E_d$, calculate E_w by Equation 5.

7a. If $m_o < E_w$, calculate k_w by Equations 7a and 7b.
 b. Calculate m by Equation 9.

8. If $E_d \geq m_o \geq E_w$, let $m = m_o$.

9. Calculate F from m by Equation 10. This is today's FFMC.

There are two restrictions in the use of these equations: 1) Equation 3 (a or b) must not be used when $r_o \leq 0.5$ mm; that is, in dry weather the rainfall routine must be omitted. 2) m has an upper limit of 250; that is, when Equation 3 (a or b) yields $m_r > 250$, let $m_r = 250$.

DUFF MOISTURE CODE (DMC)

$r_e = 0.92r_o - 1.27,$	$r_o > 1.5$	(11)
$M_o = 20 + e^{(5.6348 - P_o/43.43)}$		(12)
$b = 100/(0.5 + 0.3\,P_o),$	$P_o \leq 33$	(13a)
$b = 14 - 1.3\ln P_o,$	$33 < P_o \leq 65$	(13b)
$b = 6.2\ln P_o - 17.2,$	$P_o > 65$	(13c)
$M_r = M_o + 1000r_e/(48.77 + br_e)$		(14)
$P_r = 244.72 - 43.43\ln(M_r - 20)$		(15)
$K = 1.894\,(T + 1.1)\,(100 - H)\,L_e \times 10^{-6}$		(16)
$P = P_o\ (\text{or } P_r) + 100K$		(17)

The DMC is calculated as follows:

1. Previous day's P becomes P_o.

2a. If $r_o > 1.5$, calculate r_e by Equation 11.
 b. Calculate M_o from P_o by Equation 12.
 c. Calculate b by the appropriate one of Equations 13a, 13b, or 13c.
 d. Calculate M_r by Equation 14.
 e. Convert M_r to P_r by Equation 15. P_r becomes new P_o.

3. Take L_e from Table 1 below.

4. Calculate K by Equation 16.

5. Calculate P from P_o (or P_r) by Equation 17. This is today's DMC.

There are three restrictions on the use of the DMC equations: 1) Equations 11 to 15 are not used unless $r_o > 1.5$; that is, the rainfall routine must be omitted in dry weather. 2) P_r cannot theoretically be less than zero. Negative values resulting from Step 2e above must be raised to zero. 3) Values of T less than -1.1 must not be used in Equation 15. If $T < -1.1$, let $T = -1.1$.

Table 1
Effective daylengths (L_e) for DMC

Month:	Jan	Feb	Mar	Apr	May	June	July	Aug	Sept	Oct	Nov	Dec
L_e:	6.5	7.5	9.0	12.8	13.9	13.9	12.4	10.9	9.4	8.0	7.0	6.0

DROUGHT CODE (DC)

$r_d = 0.83r_o - 1.27,$	$r > 2.8$	(18)
$Q_o = 800e^{-D_o/400}$		(19)
$Q_r = Q_o + 3.937r_d$		(20)
$D_r = 400\ln(800/Q_r)$		(21)
$V = 0.36\,(T + 2.8) + L_f$		(22)
$D = D_o\ (\text{or } D_r) + 0.5V$		(23)

The DC is calculated as follows:

1. Previous day's D becomes D_o.

2a. If $r_o > 2.8$, calculate r_d by Equation 18.
 b. Calculate Q_o from D_o by Equation 19.
 c. Calculate Q_r by Equation 20.
 d. Convert Q_r to D_r by Equation 21. D_r becomes new D_o.

3. Take L_f from Table 2 below.

4. Calculate V by Equation 22.

5. Calculate D from D_o (or D_r) by Equation 23. This is today's DC.

There are four restrictions on the use of the DC equations: 1) Equations 18 to 21 are not used unless $r_o > 2.8$; that is, in dry weather the rainfall routine must be omitted. 2) D_r cannot theoretically be less than zero. Negative values resulting from Step 2d above must be raised to zero. 3) Values of T less than -2.8 must not be used in Equation 22. If $T < -2.8$, let $T = -2.8$. 4) V cannot be negative. If Equation 22 produces a negative result, let $V = 0$.

Table 2
Daylength factors (L_f) for DC

Month:	Jan	Feb	Mar	Apr	May	June	July	Aug	Sept	Oct	Nov	Dec
L_f:	−1.6	−1.6	−1.6	0.9	3.8	5.8	6.4	5.0	2.4	0.4	−1.6	−1.6

INITIAL SPREAD INDEX (ISI)

$$f(W) = e^{0.05039W} \quad (24)$$

$$f(F) = 91.9e^{-0.1386m}[1 + m^{5.31}/(4.93 \times 10^7)] \quad (25)$$

$$R = 0.208\ f(W)\ f(F) \quad (26)$$

BUILDUP INDEX (BUI)

$$U = 0.8\ PD/(P + 0.4D), \qquad P \leq 0.4D \quad (27a)$$

$$U = P - [1 - 0.8D/(P + 0.4D)][0.92 + (0.0114P)^{1.7}], \qquad P > 0.4D \quad (27b)$$

FIRE WEATHER INDEX (FWI)

$$f(D) = 0.626U^{0.809} + 2, \qquad U \leq 80 \quad (28a)$$

$$f(D) = 1000/(25 + 108.64\ e^{-0.023U}), \qquad U > 80 \quad (28b)$$

$$B = 0.1\ R\ f(D) \quad (29)$$

$$\ln S = 2.72\ (0.434\ \ln B)^{0.647}, \qquad B > 1 \quad (30a)$$

$$S = B, \qquad B \leq 1 \quad (30b)$$

The ISI, BUI, and FWI are calculated as follows:

1. Calculate f(W) and f(F) by Equations 24 and 25.

2. Calculate R by Equation 26. This is today's ISI.

3. Calculate U by Equation 27a if $P \leq 0.4D$, or by Equation 27b if $P > 0.4D$. This is today's BUI.

4. Calculate f(D) by Equation 28a for values of U up to 80. If $U > 80$, use Equation 28b.

5. Calculate B by Equation 29.

6. If $B > 1$, calculate S from its logarithm, given by Equation 30a. If $B \leq 1$, let $S = B$ according to Equation 30b. S is today's FWI.

DAILY SEVERITY RATING (DSR)

$$DSR = 0.0272\,(FWI)^{1.77} \quad (31)$$

DESCRIPTION OF PROGRAM

The computer program presented here descends from one written by Simard (1970) at the former Forest Fire Research Institute. That program was designed to process weather data from many stations for a whole season. There was provision for missing data and for limits to guard against abnormal weather data. It was a complex program requiring, at the time, considerable computer capacity.

Simard's program was simplified by Engisch and Walker (1971) to handle one season's verified weather data for one station only, eliminating all unnecessary limits. Kean (1975) subsequently revised this program, incorporating several newly developed mathematical changes. However, the decision to convert to metric weather measurements rendered his version obsolete. Information Report PS-X-58 (Van Wagner and Pickett 1975) included the metric successor to Kean's program. The program listed in this report incorporates the mathematical changes described earlier, but is similar in usage to the one in PS-X-58.

To operate this program, enter first the starting month and the number of days of data in that month. For example, if the starting month is April and the first day is April 13, there will be data for 18 days in month 4, and the entry will be 418. Then enter the daily weather observations in order: temperature (°C), relative humidity (%), wind (km/h), and rain (mm). Temperature and rain are entered to the first decimal place; relative humidity and wind are entered in whole numbers. The option of entering weather data in Imperial units is provided, but the output converts them to metric.

The standard starting values of the three fuel moisture codes, namely FFMC 85, DMC 6, and DC 15, are incorporated in the program. However, the option of entering any desired starting values is also provided.

The output is listed in twelve columns: date (month and day), the four kinds of weather data, the three fuel moisture codes (FFMC, DMC, DC), the three fire behavior indexes (ISI, BUI, FWI), and the Daily Severity Rating (DSR). The computer stops automatically after the last day's weather. The program includes no limits on the input data other than those needed to prevent mathematical anomalies. All moisture codes are carried from day to day in precise floating-point format, as are all intermediate quantities leading to the daily FWI. The output is thus in the purest possible mathematical form and constitutes a reference against which calculations from other computer programs or from the FWI System Tables (Canadian Forestry Service 1984) may be compared.

The main uses of this standard program[1] are for development work on the FWI System, and for reference purposes as described previously. Operational programs designed for processing daily weather observations from many stations throughout a fire season will naturally have different input and output arrangements.

Two points warrant comment. The first is the degree of precision necessary when printing out the daily codes and indexes. The standard program rounds all codes and indexes to the nearest tenth, and prints them out to one decimal place. However, whole-number (i.e., integer) output is, no doubt, adequate for most operational purposes. In certain cases, the first decimal may be desirable for the FFMC (especially above 90), and for the ISI (especialy below 10). The DSR is given to two decimal places, and is best used in that form.

The second comment refers to the precision with which the three moisture codes (FFMC, DMC, and DC) are carried over from day to day. The standard program carries over each day's code values in full floating-point format, equivalent to about seven significant figures (four or five decimal places). However, if computer storage capacity is a limiting factor, something less than perfect accuracy in output may be acceptable in operational use. Trials were made at three reduced levels of the precision to which the three moisture codes are carried over from day to day: 1) two decimals, 2) one decimal, and 3) integers. The results of two seasons' tests, as percentage of error-days in the output of each code and index, are listed in Table 3.

[1]Designated Program F-32 (August 30, 1984). Enquiries may be directed to the Petawawa National Forestry Institute, Canadian Forestry Service, Chalk River, Ontario K0J 1J0

Table 3
Percent of days in which whole-number output of the FWI System components is in error at three levels of precision in the carryover of moisture codes from day to day. Compared with standard floating-point precision.

Component	2-decimal carryover	1-decimal carryover	Integer carryover (no decimals)			
	% of days in error by 1	% of days in error by 1	% of days in error by 1	2	3	1 or more
FFMC	—	1.9	6.2	—	—	6.2
DMC	0.5	4.3	41.8	4.9	—	46.7
DC	8.1	5.2	41.3	1.4	1.6	44.3
ISI	—	0.5	3.3	—	—	3.3
BUI	—	3.5	36.1	8.2	1.4	45.7
FWI	—	0.5	9.5	—	—	9.5

Conclusions to be drawn from Table 3 are:

1. Carrying over the moisture codes with two decimal places provides nearly perfect accuracy of output except for occasional one-integer errors in the DC, which are of little consequence.
2. Carrying over one decimal place may be precise enough for some operational use. For example, one-integer errors would occur in the DMC on about one day in 25, in the FFMC on about one day in 50, and in the FWI on about one day in 200.
3. Carrying over the moisture codes as integers produces a generally inaccurate output, and is not recommended.

Of all the six components, the DC is the only one to exhibit an anomalous pattern in Table 3. Because the DC has limited power of self-correction in the mathematical sense compared with the FFMC and DMC, it tends to slip out of tune by one digit and to remain so for days on end. Considerably more sampling would be needed to establish its average pattern.

The standard program for the FWI System is listed in the following pages, followed by samples of input and output.

SYMBOLS IN THE PROGRAM

WEATHER

- T – temperature, °C
- H – relative humidity, %
- W – wind, km/h
- R,RA – rain, mm

FINE FUEL MOISTURE CODE (FFMC)

- FO – starting or yesterday's FFMC
- WMO – starting moisture content
- WM – today's final moisture content
- ED – EMC for drying
- EW – EMC for wetting
- Z – intermediate value of X
- X – log drying or wetting rate
- FFM – today's FFMC

DUFF MOISTURE CODE (DMC)

- PO – starting or yesterday's DMC
- RK – drying factor
- EL(J) – effective daylength for month J
- RW – effective rain
- PR – DMC after rain
- WMI – initial moisture content
- WMR – moisture content after rain
- B – function in rain effect

DROUGHT CODE (DC)

- DOT – starting or yesterday's DC
- PE – drying factor
- FL(J) – daylength factor for month J
- RW – effective rain
- SMI – moisture equivalent of DC
- DR – DC after rain

FIRE BEHAVIOR INDEXES (ISI, BUI, FWI)

- FM – today's fine fuel moisture content
- SF – fine fuel moisture function
- SI – ISI
- P – ratio function to correct BUI when less than DMC
- CC – DMC function to correct BUI when less than DMC
- BB – intermediate form of FWI
- SL – logarithm of final FWI
- DSR – Daily Severity Rating

```
C       PROGRAM NO.: F-32
C
C       STANDARD 1984 VERSION OF CANADIAN FOREST FIRE WEATHER INDEX SYSTEM
C       AS OF 30-AUGUST-1984.
C
C       WRITTEN IN FORTRAN 77 FOR DEC-PDP-11/44 AT PNFI.
C
C       READS WEATHER DATA IN EITHER ENGLISH OR METRIC UNITS AND PRINTS
C       OUT IN METRIC ONLY.
C
C       CODES AND INDEXES OUTPUT ALL TO ONE DECIMAL PLACE.
C       DAILY SEVERITY RATING OUTPUT TO TWO DECIMAL PLACES.
C
C       LMON = LENGTH OF MONTHS
C       EL = DMC DAY LENGTH FACTORS
C       FL = DC DAY LENGTH FACTORS
C
        DIMENSION LMON(12), EL(12), FL(12),AST(2),TITLE(20)
        LOGICAL*1 DAT(9),YES,YES1,ANS,ANS1
        LOGICAL *1 INFMT(40)
        TYPE 5
      5 FORMAT(' INPUT FILENAME: eg: SAMPLE.DAT  ',$)
        ACCEPT 10, INFMT
     10 FORMAT(40A1)
        INFMT(40)=0
        OPEN(UNIT=1,NAME=INFMT,TYPE='OLD',READONLY)
        OPEN(UNIT=2,STATUS='NEW',NAME='F32OUT.DAT')
        DATA LMON /31,28,31,30,31,30,31,31,30,31,30,31/
        DATA EL /6.5,7.5,9.0,12.8,13.9,13.9,12.4,10.9,9.4,8.0,7.0,
       *6.0/
        DATA FL /-1.6,-1.6,-1.6,.9,3.8,5.8,6.4,5.0,2.4,.4,-1.6,-1.6/
        DATA AST /' ','*'/
        DATA ANS /'Y'/
        TYPE 15
     15 FORMAT(//1X,'ENTER VIA KEYBOARD EITHER 0 OR 1 '/1X'......0
       * IF DATA IS IN METRIC UNITS'/1X' OR 1 IF DATA IS IN
       * ENGLISH UNITS')
        READ(5,*) IUNIT
C
C       20 - METRIC FORMAT, 25 - ENGLISH FORMAT.
C
     20 FORMAT(F4.1,2I4,F5.1)
     25 FORMAT(F4.0,2I4,F5.2)
C
C       READS IN STATION & YEAR.
C
        READ(1,30) TITLE
     30 FORMAT(20A4)
C
C       THIS SECTION ALLOWS FOR INITIAL VALUES OF FFMC, DMC, DC TO BE OPTIONAL.
C
        FO=85.0
        PO=6.0
        DOT=15.0
        TYPE 35
     35 FORMAT(' FFMC=85.0, DMC=6.0, DC=15.0; DO YOU WISH TO USE THESE
       * INITIAL STANDARD VALUES? [Y/N] '$)
        ACCEPT 10,ANS1
        IF(ANS1.EQ.ANS) GO TO 55
        TYPE 40
```

```
   40 FORMAT(' FFMC (F4.1): ',$)
      ACCEPT 20,FO
      TYPE 45
   45 FORMAT(' DMC (F4.1): ',$)
      ACCEPT 20,PO
      TYPE 50
   50 FORMAT(' DC (F4.1): ',$)
      ACCEPT 20,DOT
C
C     READS STARTING MONTH OF THE YEAR AND NUMBER OF DAYS IN STARTING
C     MONTH.
C
   55 READ(1,60)  M, NDAYS
   60 FORMAT(I1,I2)
      WRITE(2,65)
   65 FORMAT(1H1'PROGRAM NO.: F-32')
      CALL DATE(DAT)
      WRITE(2,70) DAT,TITLE
   70 FORMAT(1X,9A1///1X,20A4//)
      WRITE(2,75) FO,PO,DOT
   75 FORMAT(' INITIAL VALUES FOR FFMC:',F5.1,', DMC:',F5.1,', DC:',F5.1,
     *//)
      DO 290 J=M,12
      NN=LMON(J)
      IF(J.EQ.M) GO TO 80
      IDAYS=1
      GO TO 85
   80 IDAYS=LMON(J)-NDAYS+1
C
C     READS DAILY WEATHER DATA
C
   85 IAST=1
      DO 290 I=IDAYS,NN
      IF(IUNIT.EQ.0) GO TO 90
      READ(1,25,END=295) T,IH,IW,R
      W=IW
      GO TO 95
   90 READ(1,20,END=295) T,IH,IW,R
      W=IW
   95 IF(IUNIT.EQ.0) GO TO 100
      T=(T-32.)*5./9.
      W=W*1.609
      R=R*25.4
  100 TX=T
      H=IH
```

```
      RAIN=R
      IF(IAST.GT.1) GO TO 110
      WRITE(2,105)
  105 FORMAT(///,1X,' DATE  TEMP  RH   WIND  RAIN
     *   FFMC   DMC    DC
     *    ISI   BUI   FWI      DSR'/)
C
C     FINE FUEL MOISTURE CODE
C
  110 WMO=(147.2*(101-FO))/(59.5+FO)
      IF(R.GT.0.5) GO TO 115
      GO TO 125
  115 RA=R-.5
      IF(WMO.GT.150.) GO TO 120
      WMO=WMO+42.5*RA*EXP(-100./(251.-WMO))*(1.-EXP(-6.93/RA))
      GO TO 125
  120 WMO=(WMO+42.5*RA*EXP(-100./(251.-WMO))*(1.-EXP(-6.93/RA)))+
     *(0.0015*(WMO-150.)**2)*RA**.5
  125 IF(WMO.GT.250.) WMO=250.
      ED=0.942*(H**0.679)+(11.*EXP((H-100.)/10.))+0.18*(21.1-T)
     **(1.-1./EXP(0.115*H))
      IF(WMO-ED) 130,135,140
  130 EW=.618*(H**.753)+(10.*EXP((H-100.)/10.))+.18*(21.1-T)
     **(1.-1./EXP(0.115*H))
      IF(WMO.LT.EW) GO TO 145
  135 WM=WMO
      GO TO 150
  140 Z=0.424*(1.-(H/100.)**1.7)+(0.0694*(W**0.5))*(1.-(H/100.)**8)
      X=Z*(0.581*(EXP(0.0365*T)))
      WM=ED+(WMO-ED)/10.**X
      GO TO 150
  145 Z=.424*(1.-((100.-H)/100.)**1.7)+(.0694*(W**.5))*(1.-((100.-H)
     */100.)**8)
      X=Z*(.581*(EXP(.0365*T)))
      WM=EW-(EW-WMO)/10.**X
  150 FFM=(59.5*(250.-WM))/(147.2+WM)
      IF(FFM.GT.101.) GO TO 155
      IF(FFM) 160,165,165
  155 FFM=101.
      GO TO 165
  160 FFM=0.0
C
C     DUFF MOISTURE CODE
C
  165 IF(T+1.1.GE.0.) GO TO 170
      T=-1.1
  170 RK=1.894*(T+1.1)*(100.-H)*(EL(J)*0.0001)
  175 IF(R.GT.1.5) GO TO 180
      PR=PO
      GO TO 205
  180 RA=R
      RW=0.92*RA-1.27
      WMI=20.0+280./EXP(0.023*PO)
      IF(PO.LE.33.) GO TO 185
      IF(PO-65.) 190,190,195
  185 B=100./(0.5+0.3*PO)
      GO TO 200
  190 B=14.-1.3*ALOG(PO)
      GO TO 200
  195 B=6.2*ALOG(PO)-17.2
  200 WMR=WMI+(1000.*RW)/(48.77+B*RW)
      PR=43.43*(5.6348-ALOG(WMR-20.))
  205 IF(PR.GE.0.) GO TO 210
      PR=0.0
  210 DMC=PR+RK
C
```

```
C     DROUGHT CODE
C
      IF(T+2.8.GE.0.) GO TO 215
      T=-2.8
  215 PE=(.36*(T+2.8)+FL(J))/2.
      IF(R.LE.2.8) GO TO 225
      RA=R
      RW=0.83*RA-1.27
      SMI=800.*EXP(-DOT/400.)
      DR=DOT-400.*ALOG(1.+((3.937*RW)/SMI))
      IF(DR.GT.0.) GO TO 220
      DR=0.0
  220 DC=DR+PE
      GO TO 230
  225 DR=DOT
      GO TO 220
  230 IF(DC.GE.0.) GO TO 235
      DC=0.0
C
C     INITIAL SPREAD INDEX, BUILDUP INDEX, FIRE WEATHER INDEX
C
  235 FM=(147.2*(101.-FFM))/(59.5+FFM)
      SF=19.115*EXP(FM*(-.1386))*(1.+(FM**5.31)/4.93E07)
      SI=SF*EXP(0.05039*W)
  240 IF(DMC.EQ.0.0.AND.DC.EQ.0.0) GO TO 245
      BUI=(0.8*DC*DMC)/(DMC+0.4*DC)
      GO TO 250
  245 BUI=0.
  250 IF(BUI.GE.DMC) GO TO 255
      P=(DMC-BUI)/DMC
      CC=0.92+(0.0114*DMC)**1.7
      BUI=DMC-(CC*P)
      IF(BUI.LT.0.) BUI=0.
  255 IF(BUI.GT.80.) GO TO 260
      BB=0.1*SI*(0.626*BUI**0.809+2.)
      GO TO 265
  260 BB=0.1*SI*(1000./(25.+108.64/EXP(0.023*BUI)))
  265 IF(BB-1.0.LE.0.) GO TO 270
      SL=2.72*(0.434*ALOG(BB))**0.647
      FWI=EXP(SL)
      GO TO 275
  270 FWI=BB
  275 DSR = 0.0272*FWI**1.77
  280 WRITE(2,285) J,I,TX,IH,IW,RAIN,FFM,DMC,DC,SI,BUI,FWI,DSR
  285 FORMAT(1X,2I3,F6.1,I4,I6,F7.1,F7.1,F6.1,F7.1,3F6.1,F8.2)
      FO=FFM
      PO=DMC
      DOT=DC
      IAST=IAST+1
  290 CONTINUE
  295 WRITE(5,300)
  300 FORMAT(/1X'OUTPUT IN F32OUT.DAT AND MUST BE SPOOLED
     * TO THE LINE PRINTER.'//)
  305 STOP
      END
```

SAMPLE OF INPUT

```
FWI TEST DATA APRIL 13 TO MAY 31
418
 17.  42  25               – April 13
 20.  21  25 2.4
 8.5  40  17
 6.5  25   6
 13.  34  24
 6.0  40  22 0.4
 5.5  52   6
 8.5  46  16
 9.5  54  20
 7.0  93  14 9.0
 6.5  71  17 1.0
 6.0  59  17
 13.  52   4
15.5  40  11
 23.  25   9
 19.  46  16               – April 30
 18.  41  20               – May 1
14.5  51  16
14.5  69  11
15.5  42   8
 21.  37   8
 23.  32  16
 23.  32  14
 27.  33  12
 28.  17  27
23.5  54  20
 16.  50  2212.2
11.5  58  20
 16.  54  16
21.5  37   9
 14.  61  22 0.2
 15.  30  27
 20.  23  11
 14.  95   316.4
 20.  53   4 2.8
19.5  30  16
25.5  51  20 6.0
 10.  38  24
 19.  27  16
 26.  46  11 4.2
 30.  38  22
25.5  67  1912.6
 12.  53  2811.8
 21.  38   8
 13.  70  20 3.8
  9.  78  24 1.4
 11.  54  16
15.5  39   9
 18.  36   5               – May 31
```

4 indicates starting month, in this case April.
18 indicates no. of days in starting month.

Temperature °C | Relative Humidity % | Wind km/h | Rain mm

SAMPLE OF OUTPUT

PROGRAM NO.: F-32
12-FEB-85

FWI TEST DATA APRIL 13 TO MAY 31

INITIAL VALUES FOR FFMC: 85.0, DMC: 6.0, DC: 15.0

DATE	TEMP	RH	WIND	RAIN	FFMC	DMC	DC	ISI	BUI	FWI	DSR
4 13	17.0	42	25	0.0	87.7	8.5	19.0	10.9	8.5	10.1	1.63
4 14	20.0	21	25	2.4	86.2	10.4	23.6	8.8	10.4	9.3	1.40
4 15	8.5	40	17	0.0	87.0	11.8	26.1	6.5	11.7	7.6	0.98
4 16	6.5	25	6	0.0	88.8	13.2	28.2	4.9	13.1	6.2	0.68
4 17	13.0	34	24	0.0	89.1	15.4	31.5	12.6	15.3	14.8	3.22
4 18	6.0	40	22	0.4	88.7	16.5	33.5	10.7	16.4	13.5	2.71
4 19	5.5	52	6	0.0	87.4	17.2	35.4	4.0	17.1	5.9	0.62
4 20	8.5	46	16	0.0	87.4	18.5	37.9	6.6	18.4	9.7	1.50
4 21	9.5	54	20	0.0	86.8	19.7	40.6	7.4	19.6	11.0	1.89
4 22	7.0	93	14	9.0	29.9	10.1	29.5	0.0	10.9	0.0	0.00
4 23	6.5	71	17	1.0	49.4	10.7	31.6	0.4	11.6	0.2	0.00
4 24	6.0	59	17	0.0	67.3	11.4	33.7	1.3	12.3	0.9	0.02
4 25	13.0	52	4	0.0	77.8	13.0	37.0	1.1	13.9	0.8	0.02
4 26	15.5	40	11	0.0	85.5	15.4	40.7	3.9	15.9	5.5	0.56
4 27	23.0	25	9	0.0	91.5	19.8	45.8	8.4	19.8	12.2	2.27
4 28	19.0	46	16	0.0	89.9	22.5	50.2	9.5	22.4	14.3	3.02
4 29	18.0	41	20	0.0	90.0	25.2	54.4	11.7	25.1	17.7	4.40
4 30	14.5	51	16	0.0	88.4	27.0	57.9	7.7	27.0	13.3	2.65
5 1	14.5	69	11	0.0	85.7	28.3	63.0	4.0	28.2	8.0	1.08
5 2	15.5	42	8	0.0	87.4	30.8	68.2	4.4	30.8	9.1	1.35
5 3	21.0	37	8	0.0	89.4	34.5	74.3	5.9	34.4	12.3	2.30
5 4	23.0	32	16	0.0	91.0	38.8	80.9	11.1	38.7	21.0	5.96
5 5	23.0	32	14	0.0	91.2	43.1	87.4	10.3	43.0	21.1	5.99
5 6	27.0	33	12	0.0	91.7	48.1	94.7	9.9	47.9	21.7	6.30
5 7	28.0	17	27	0.0	95.2	54.5	102.1	34.5	54.3	52.6	30.20
5 8	23.5	54	20	0.0	89.7	57.4	108.8	11.3	57.2	25.9	8.61
5 9	16.0	50	22	12.2	62.2	29.9	91.8	1.4	33.0	3.0	0.19
5 10	11.5	58	20	0.0	76.7	31.3	96.3	2.3	34.5	5.4	0.55
5 11	16.0	54	16	0.0	83.5	33.4	101.6	3.8	36.7	8.9	1.31
5 12	21.5	37	9	0.0	88.7	37.1	107.8	5.6	39.9	12.8	2.46
5 13	14.0	61	22	0.2	86.7	38.7	112.8	8.1	41.6	17.3	4.24
5 14	15.0	30	27	0.0	89.6	41.7	117.9	15.9	44.2	28.8	10.44
5 15	20.0	23	11	0.0	92.1	45.9	123.9	10.1	47.7	21.9	6.40
5 16	14.0	95	3	16.4	21.3	20.1	97.0	0.0	26.5	0.0	0.00
5 17	20.0	53	4	2.8	51.0	18.3	103.0	0.2	25.3	0.2	0.00
5 18	19.5	30	16	0.0	82.3	22.1	108.9	3.3	29.3	6.8	0.81
5 19	25.5	51	20	6.0	75.4	16.4	106.4	2.1	23.7	3.8	0.29
5 20	10.0	38	24	0.0	84.3	18.2	110.6	6.4	25.8	11.3	1.99
5 21	19.0	27	16	0.0	90.3	22.1	116.4	10.0	29.9	17.2	4.19
5 22	26.0	46	11	4.2	77.6	18.7	117.7	1.6	26.8	2.9	0.18
5 23	30.0	38	22	0.0	90.2	23.8	125.5	13.4	32.3	22.0	6.49
5 24	25.5	67	19	12.6	65.3	13.1	108.5	1.4	20.2	1.9	0.08
5 25	12.0	53	28	11.8	55.4	7.7	91.6	1.2	12.8	0.8	0.02
5 26	21.0	38	8	0.0	80.8	11.3	97.8	1.9	17.6	2.6	0.14
5 27	13.0	70	20	3.8	61.7	8.4	97.9	1.2	13.8	0.9	0.02
5 28	9.0	78	24	1.4	64.5	9.0	101.9	1.7	14.7	2.0	0.09
5 29	11.0	54	16	0.0	77.6	10.5	106.3	2.0	16.8	2.8	0.17
5 30	15.5	39	9	0.0	85.4	13.1	111.5	3.5	20.3	5.8	0.61
5 31	18.0	36	5	0.0	88.5	16.3	117.1	4.4	24.2	7.9	1.06

REFERENCES

Canadian Forestry Service. 1984. Tables for the Canadian Forest Fire Weather Index System. Environ. Can., Can. For. Serv., For. Tech. Rep. 25 (4th ed.). 48 p.

Engisch, R.L.; Walker, J.D. 1971. PDP-8L version of Simard's Fire Weather Index Program. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta., Intern. Rep. PS-23. 10 p.

Kean, W.A. 1975. A PDP-8L program for calculating the Fire Weather Index. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta., Inf. Rep. PS-X-57. 12 p.

Simard, A.J. 1970. Computer program to calculate the Canadian Forest Fire Weather Index. Environ. Can., Can. For. Serv., Forest Fire Res. Inst., Intern. Rep. FF-12. 18 p.

Van Wagner, C.E. 1970. Conversion of Williams' severity rating for use with the Fire Weather Index. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta., Inf. Rep. PS-X-21. 5 p.

Van Wagner, C.E. 1974. Structure of the Canadian Forest Fire Weather Index. Environ. Can., Can. For. Serv., Pub. No. 1333. 44 p.

Van Wagner, C.E. 1977. A method of computing fine fuel moisture content throughout the diurnal cycle. Environ. Can., Can. For. Serv., Petawawa Forest Exp. Sta. Inf. Rep. PS-X-69. 15 p.

Van Wagner, C.E.; Pickett, T.L. 1975. Equations and Fortran IV program for the 1976 metric version of the Forest Fire Weather Index. Can. For. Serv., Petawawa Forest Exp. Sta., Inf. Rep. PS-X-58. 20 p.